

Time-Invariant Predictors of Practice Effects

The models for this example use the same response time data as in Hoffman (in preparation) chapter 6. We will be examining baseline age, abstract reasoning, and education level as time-invariant predictors of change in response time (RT) in milliseconds over six practice sessions to a measure of processing speed (as measured by the number match 3 test) in a sample of 101 older adults.

SAS Syntax for Data Manipulation:

```
* Location for original SAS files for these models - change this path;
%LET example = F:\Example Data\Chapter 6 Data\CHAP Burst 1 RT;
LIBNAME example "&example.";
* Defining macro variable for datafile name to be replaced in code below;
%LET datafile=Example7a;

* Bringing data into work library and recoding/centering variables;
* Centering time at different points for polynomial models;
DATA &datafile.; SET example.&datafile.;
    * Centering time at session 1 or 6 for polynomial models;
    c1sess = session - 1; c6sess = session - 6;
    LABEL c1sess = "Session (0=1)" c6sess = "Session (0=6)";
* Creating two slopes for piecewise models;
    IF Session = 1 THEN DO; Slope12 = 0; Slope26 = 0; END;
    ELSE IF Session = 2 THEN DO; Slope12 = 1; Slope26 = 0; END;
    ELSE IF Session > 2 THEN DO; Slope12 = 1; Slope26 = Session-2; END;
    LABEL Slope12 = "1-2 Early Practice Slope"
           Slope26 = "2-6 Later Practice Slope";
* Centering level-2 predictors;
Age80 = baseage - 80;
Reas22 = AbsReas - 22;
LABEL Age80 = "Age Centered (0=80)"
       Reas22 = "Abstract Reasoning Centered (0=22)";
* Make education a grouping variable FOR DEMO PURPOSES;
    IF EducYrs = . THEN EducGrp = .;
    ELSE IF EducYrs LE 12 THEN EducGrp=1;
    ELSE IF EducYrs GT 12 AND EducYrs LE 16 THEN EducGrp=2;
    ELSE IF EducYrs GT 16 THEN EducGrp=3;
    LABEL EducGrp= "Education Group (1=HS, 2=BA, 3=GRAD)"; RUN;

* REMOVING CASES WITH MISSING PREDICTORS OR OUTCOME;
DATA trimmed; SET &datafile.;
    WHERE NMISS(Age80, Reas22, EducGrp, session, nm3rt)=0; RUN;
* Changing dataset used in analyses below;
%LET datafile=trimmed;
```

Model 1a. Baseline Unconditional Random Piecewise Growth Model in ML

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}(\text{Slope12}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + U_{0i}$$

$$\text{Slope12: } \beta_{1i} = \gamma_{10} + U_{1i}$$

$$\text{Slope26: } \beta_{2i} = \gamma_{20} + U_{2i}$$

```
TITLE1 "Piecewise Unconditional Model - Random Early/Later Practice Slopes";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST IC NAMELEN=100 METHOD=ML;
    CLASS ID session;
    MODEL nm3rt = Slope12 Slope26
        / SOLUTION DDFM=Satterthwaite OUTPM=PredPiece; *Save time-predicted RT;;
    RANDOM INTERCEPT Slope12 Slope26 / TYPE=UN SUBJECT=ID;
    REPEATED session / TYPE=VC SUBJECT=ID;
    ODS OUTPUT CovParms=CovPiece InfoCrit=FitPiece; * Save covparms, fit for comparison;
RUN; PROC CORR DATA=PredPiece; VAR pred nm3rt; RUN; * Correlation of predicted and actual RT;
```

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	
UN(1,1)	ID	281322	42099	6.68	<.0001	Random Intercept variance
UN(2,1)	ID	-53558	17962	-2.98	0.0029	Int-Slope12 covariance
UN(2,2)	ID	63041	13054	4.83	<.0001	Random Slope12 variance
UN(3,1)	ID	-10538	3735.10	-2.82	0.0048	Int-Slope26 covariance
UN(3,2)	ID	-1620.74	2066.55	-0.78	0.4329	Slope12-Slope26 covariance
UN(3,3)	ID	2573.86	627.53	4.10	<.0001	Random Slope26 variance
Session	ID	17673	1435.84	12.31	<.0001	Residual (e) variance

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8298.9	10	8318.9	8319.3	8329.5	8345.1	8355.1

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t	
Intercept	1961.89	54.4091	101	36.06	<.0001	Predicted RT when time=0 (session 1 here)
Slope12	-163.64	30.0689	101	-5.44	<.0001	RT Change/session between sessions 1 and 2
Slope26	-32.8932	6.5561	101	-5.02	<.0001	RT Change/session between sessions 2 and 6

Pearson Correlation Coefficients, N = 606

Prob > |r| under H0: Rho=0

	Pred	nm3rt
Pred	1.00000	0.19338
Predicted Mean		<.0001

$r = .1938 \rightarrow \text{TOTAL } R^2 = .0376$
 ~ 4% of RT variance is accounted for
 by 2 piecewise effects of session

Model 1b. Piecewise Model with Fixed Effects of Age on Intercept, Slope12, and Slope26

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}(\text{Slope12}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01}(\text{Age}_i - 80) + U_{0i}$$

$$\text{Slope12: } \beta_{1i} = \gamma_{10} + \gamma_{11}(\text{Age}_i - 80) + U_{1i}$$

$$\text{Slope26: } \beta_{2i} = \gamma_{20} + \gamma_{21}(\text{Age}_i - 80) + U_{2i}$$

```
TITLE1 "Add Fixed Effects for Age on Intercept, Slope12, and Slope26";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST IC NAMELEN=100 METHOD=ML;
  CLASS ID session;
  MODEL nm3rt = Slope12 Slope26 Age80 Age80*Slope12 Age80*Slope26
    / SOLUTION DDFM=Satterthwaite OUTPM=PredAge; * Save fixed-predicted RT;
  RANDOM INTERCEPT Slope12 Slope26 / G TYPE=UN SUBJECT=ID;
  REPEATED session / TYPE=VC SUBJECT=ID;
  ODS OUTPUT CovParms=CovAge InfoCrit=FitAge; * Save covparms, fit for comparison;
  * Requesting slopes at age 90;
  ESTIMATE "Slope 1-2 for Age 80" Slope12 1 Age80*Slope12 0;
  ESTIMATE "Slope 1-2 for Age 90" Slope12 1 Age80*Slope12 10;
  ESTIMATE "Slope 2-6 for Age 80" Slope26 1 Age80*Slope26 0;
  ESTIMATE "Slope 2-6 for Age 90" Slope26 1 Age80*Slope26 10;
  * Requesting additional effects for age;
  ESTIMATE "Age Effect at Session 1" Age80 1 Age80*Slope12 0 Age80*Slope26 0;
  ESTIMATE "Age Effect at Session 2" Age80 1 Age80*Slope12 1 Age80*Slope26 0;
  ESTIMATE "Age Effect at Session 3" Age80 1 Age80*Slope12 1 Age80*Slope26 1;
  ESTIMATE "Age Effect at Session 4" Age80 1 Age80*Slope12 1 Age80*Slope26 2;
  ESTIMATE "Age Effect at Session 5" Age80 1 Age80*Slope12 1 Age80*Slope26 3;
  ESTIMATE "Age Effect at Session 6" Age80 1 Age80*Slope12 1 Age80*Slope26 4;
RUN;
```

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	ID	248900	37540	6.63	<.0001
UN(2,1)	ID	-45304	16637	-2.72	0.0065
UN(2,2)	ID	60940	12763	4.77	<.0001
UN(3,1)	ID	-9068.73	3477.97	-2.61	0.0091
UN(3,2)	ID	-1994.87	2039.86	-0.98	0.3281
UN(3,3)	ID	2507.25	618.41	4.05	<.0001
session	ID	17673	1435.84	12.31	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8287.2	13	8313.2	8313.8	8326.9	8347.2	8360.2

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	1966.86	51.3940	101	38.27	<.0001
Slope12	-164.91	29.7322	101	-5.55	<.0001
Slope26	-33.1182	6.5080	101	-5.09	<.0001
Age80	29.7804	8.4968	101	3.50	0.0007
Slope12*Age80	-7.5810	4.9156	101	-1.54	0.1261
Slope26*Age80	-1.3499	1.0760	101	-1.25	0.2125

Estimates						
Label	Estimate	Standard Error	DF	t Value	Pr > t	
Slope 1-2 for Age 80	-164.91	29.7322	101	-5.55	<.0001	
Slope 1-2 for Age 90	-240.72	58.1448	101	-4.14	<.0001	
Slope 2-6 for Age 80	-33.1182	6.5080	101	-5.09	<.0001	
Slope 2-6 for Age 90	-46.6173	12.7272	101	-3.66	0.0004	
Age Effect at Session 1	29.7804	8.4968	101	3.50	0.0007	
Age Effect at Session 2	22.1993	7.8896	101	2.81	0.0059	
Age Effect at Session 3	20.8494	7.4496	101	2.80	0.0061	
Age Effect at Session 4	19.4995	7.1458	101	2.73	0.0075	
Age Effect at Session 5	18.1496	6.9960	101	2.59	0.0109	
Age Effect at Session 6	16.7997	7.0100	101	2.40	0.0184	

Is the age piecewise model (1b) better than the unconditional piecewise growth model (1a)?
How do we know?

Interpret the fixed intercept:

Interpret the fixed effect of Slope12:

Interpret the fixed effect of Slope26:

Interpret the effect of Age80:

Interpret the effect of Slope12*Age80:

Interpret the effect of Slope26*Age80:

Syntax and output for Total R² from this model:

```
PROC CORR DATA=PredAge; VAR pred nm3rt; RUN; * Correlation of predicted and actual RT;
```

Pearson Correlation Coefficients, N = 606
 Prob > |r| under H0: Rho=0

	Pred	nm3rt
Pred	1.00000	0.32795
Predicted Mean		<.0001

$r = .32795 \rightarrow \text{TOTAL } R^2 = .1076$
 ~ 11% of RT variance is accounted for
 by 2 piecewise effects of session + age
 (~7% due to unique effects of age)

Syntax and output for FitTest macro to compare -2LL across models:

```
* Calculate difference in model fit relative to unconditional model;
%FitTest(FitFewer=FitPiece, NameFewer=Unconditional, FitMore=FitAge, NameMore=Age);
```

Add Fixed Effects for Age on Intercept, Slope12, and Slope26
 Likelihood Ratio Test for Unconditional vs. Age

Name	Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
Unconditional	8298.9	10	8318.9	8345.1	.	.	.
Age	8287.2	13	8313.2	8347.2	11.7659	3	.008229493

Syntax and output for Total-R² macro to compare total R² values across models:

```
* Calculate Total R2 change relative to unconditional model;
%TotalR2(DV=nm3rt, PredFewer=PredPiece, NameFewer=Unconditional, PredMore=PredAge, NameMore=Age);
```

Add Fixed Effects for Age on Intercept, Slope12, and Slope26
 Total R2 (% Reduction) for Unconditional vs. Age

Name	Pred Corr	TotalR2	Total R2Diff
Unconditional	0.19338	0.03740	.
Age	0.32795	0.10755	0.070156

Syntax and output for Pseudo-R² macro to compare variance components across models:

```
* Calculate PseudoR2 relative to unconditional model;
%PseudoR2(NCov=7, CovFewer=CovPiece, NameFewer=Unconditional, CovMore=CovAge, NameMore=Age);
```

Add Fixed Effects for Age on Intercept, Slope12, and Slope26
 PsuedoR2 (% Reduction) for CovPiece vs. CovAge

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	Pseudo R2
Unconditional	UN(1,1)	ID	281322	42099	6.68	<.0001	.
Unconditional	UN(2,2)	ID	63041	13054	4.83	<.0001	.
Unconditional	UN(3,3)	ID	2573.86	627.53	4.10	<.0001	.
Unconditional	session	ID	17673	1435.84	12.31	<.0001	.
Age	UN(1,1)	ID	248900	37540	6.63	<.0001	0.11525
Age	UN(2,2)	ID	60940	12763	4.77	<.0001	0.03333
Age	UN(3,3)	ID	2507.25	618.41	4.05	<.0001	0.02588
Age	session	ID	17673	1435.84	12.31	<.0001	0.00000

Which variance component should have been reduced by each new fixed effect of age?

Model 1b. Piecewise Model with Fixed Effects of Age and Reasoning on Intercept, Slope12, Slope26

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i} (\text{Slope12}_{ti}) + \beta_{2i} (\text{Slope26}_{ti}) + e_{ti}$

Level 2: Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01} (\text{Age}_i - 80) + \gamma_{02} (\text{Reason}_i - 22) + U_{0i}$

Slope12: $\beta_{1i} = \gamma_{10} + \gamma_{11} (\text{Age}_i - 80) + \gamma_{12} (\text{Reason}_i - 22) + U_{1i}$

Slope26: $\beta_{2i} = \gamma_{20} + \gamma_{21} (\text{Age}_i - 80) + \gamma_{22} (\text{Reason}_i - 22) + U_{2i}$

```
TITLE1 "Add Fixed Effects for Reasoning on Intercept, Slope12, and Slope26";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST NAMELEN=100 IC METHOD=ML;
CLASS ID session;
MODEL nm3rt = Slope12 Slope26 Age80 Age80*Slope12 Age80*Slope26
          Reas22 Reas22*Slope12 Reas22*Slope26
          / SOLUTION DDFM=Satterthwaite OUTPM=PredReas; * Save fixed-predicted RT;
RANDOM INTERCEPT Slope12 Slope26 / G TYPE=UN SUBJECT=ID;
REPEATED session / TYPE=VC SUBJECT=ID;
ODS OUTPUT CovParms=CovReas InfoCrit=FitReas; * Save covparms, fit for comparison;
* Requesting additional effects for reasoning;
ESTIMATE "Reasoning Effect at Session 1" Reas22 1 Reas22*Slope12 0 Reas22*Slope26 0;
ESTIMATE "Reasoning Effect at Session 2" Reas22 1 Reas22*Slope12 1 Reas22*Slope26 0;
ESTIMATE "Reasoning Effect at Session 3" Reas22 1 Reas22*Slope12 1 Reas22*Slope26 1;
ESTIMATE "Reasoning Effect at Session 4" Reas22 1 Reas22*Slope12 1 Reas22*Slope26 2;
ESTIMATE "Reasoning Effect at Session 5" Reas22 1 Reas22*Slope12 1 Reas22*Slope26 3;
ESTIMATE "Reasoning Effect at Session 6" Reas22 1 Reas22*Slope12 1 Reas22*Slope26 4;
RUN;
```

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	ID	234473	35511	6.60	<.0001
UN(2,1)	ID	-47812	16314	-2.93	0.0034
UN(2,2)	ID	60504	12703	4.76	<.0001
UN(3,1)	ID	-7287.88	3263.20	-2.23	0.0255
UN(3,2)	ID	-1685.32	1979.00	-0.85	0.3944
UN(3,3)	ID	2287.42	588.37	3.89	<.0001
session	ID	17673	1435.84	12.31	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8276.9	16	8308.9	8309.8	8325.8	8350.7	8366.7

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	1982.64	50.4135	101	39.33	<.0001
Slope12	-162.16	29.9144	101	-5.42	<.0001
Slope26	-35.0669	6.3930	101	-5.49	<.0001
Age80	23.0041	8.7312	101	2.63	0.0097
Slope12*Age80	-8.7589	5.1810	101	-1.69	0.0940
Slope26*Age80	-0.5135	1.1072	101	-0.46	0.6438
Reas22	-27.1200	11.2814	101	-2.40	0.0180
Slope12*Reas22	-4.7141	6.6942	101	-0.70	0.4829
Slope26*Reas22	3.3476	1.4306	101	2.34	0.0213

Which fixed effects are conditional on age?

Which fixed effects are conditional on reasoning?

Label	Estimates		DF	t Value	Pr > t
	Estimate	Standard Error			
Reasoning Effect at Session 1	-27.1200	11.2814	101	-2.40	0.0180
Reasoning Effect at Session 2	-31.8340	10.2944	101	-3.09	0.0026
Reasoning Effect at Session 3	-28.4864	9.7671	101	-2.92	0.0044
Reasoning Effect at Session 4	-25.1388	9.4291	101	-2.67	0.0089
Reasoning Effect at Session 5	-21.7912	9.3014	101	-2.34	0.0211
Reasoning Effect at Session 6	-18.4436	9.3923	101	-1.96	0.0523

Syntax and output for FitTest macro to compare -2LL across models:

```
* Calculate difference in model fit relative to age model;
%FitTest(FitFewer=FitAge, NameFewer=Age, FitMore=FitReas, NameMore=Reasoning);
```

Add Fixed Effects for Age on Intercept, Slope12, and Slope26
Likelihood Ratio Test for Age vs. Reasoning

Name	Neg2Log		AIC	BIC	DevDiff	DFdiff	Pvalue
	Like	Parms					
Age	8287.2	13	8313.2	8347.2	.	.	.
Reasoning	8276.9	16	8308.9	8350.7	10.2871	3	0.016277

```
* Calculate difference in model fit relative to unconditional model;
%FitTest(FitFewer=FitPiece, NameFewer=Unconditional, FitMore=FitReas, NameMore=Reasoning);
```

Add Fixed Effects for Age on Intercept, Slope12, and Slope26
Likelihood Ratio Test for Unconditional vs. Reasoning

Name	Neg2Log		AIC	BIC	DevDiff	DFdiff	Pvalue
	Like	Parms					
Unconditional	8298.9	10	8318.9	8345.1	.	.	.
Reasoning	8276.9	16	8308.9	8350.7	22.0531	6	.001184349

Syntax and output for Total-R² macro to compare total R² values across models:

```
* Calculate Total R2 change relative to age model;
%TotalR2(DV=nm3rt, PredFewer=PredAge, NameFewer=Age, PredMore=PredReas, NameMore=Reasoning);
```

Add Fixed Effects for Age on Intercept, Slope12, and Slope26
Total R2 (% Reduction) for Age vs. Reasoning

Name	Pred		Total
	Corr	TotalR2	R2Diff
Age	0.32795	0.10755	.
Reasoning	0.40163	0.16131	0.053755

Syntax and output for Pseudo-R² macro to compare variance components across models:

```
* Calculate PseudoR2 relative to age model;
%PseudoR2(NCov=7, CovFewer=CovAge, NameFewer=Age, CovMore=CovReas, NameMore=Reasoning);
```

Add Fixed Effects for Age on Intercept, Slope12, and Slope26
PseudoR2 (% Reduction) for Age vs. Reasoning

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
Age	UN(1,1)	ID	248900	37540	6.63	<.0001	.
Age	UN(2,2)	ID	60940	12763	4.77	<.0001	.
Age	UN(3,3)	ID	2507.25	618.41	4.05	<.0001	.
Age	session	ID	17673	1435.84	12.31	<.0001	.
Reasoning	UN(1,1)	ID	234473	35511	6.60	<.0001	0.057964
Reasoning	UN(2,2)	ID	60504	12703	4.76	<.0001	0.007153
Reasoning	UN(3,3)	ID	2287.42	588.37	3.89	<.0001	0.087675
Reasoning	session	ID	17673	1435.84	12.31	<.0001	-0.000000

Model 1c. Piecewise Model Adding Education Group on Intercept, Slope12, Slope26

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Slope12}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$

Level 2:

Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01}(\text{Age}_i - 80) + \gamma_{02}(\text{Reason}_i - 22) + \gamma_{03}(\text{Highvs.LowEd}_i) + \gamma_{04}(\text{Highvs.MedEd}_i) + U_{0i}$

Slope12: $\beta_{1i} = \gamma_{10} + \gamma_{11}(\text{Age}_i - 80) + \gamma_{12}(\text{Reason}_i - 22) + \gamma_{13}(\text{Highvs.LowEd}_i) + \gamma_{14}(\text{Highvs.MedEd}_i) + U_{1i}$

Slope26: $\beta_{2i} = \gamma_{20} + \gamma_{21}(\text{Age}_i - 80) + \gamma_{22}(\text{Reason}_i - 22) + \gamma_{23}(\text{Highvs.LowEd}_i) + \gamma_{24}(\text{Highvs.MedEd}_i) + U_{2i}$

```
TITLE1 "Add Effects of Education Group on Intercept, Slope12, and Slope26";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST IC NAMELEN=100 METHOD=ML;
  CLASS ID EducGrp session;
  MODEL nm3rt = Slope12 Slope26 Age80 Age80*Slope12 Age80*Slope26
    Reas22 Reas22*Slope12 Reas22*Slope26
    EducGrp Slope12*EducGrp Slope26*EducGrp
    / SOLUTION DDFM=Satterthwaite OUTPM=PredEdPiece; * Save fixed-predicted RT;
  RANDOM INTERCEPT Slope12 Slope26 / G TYPE=UN SUBJECT=ID;
  REPEATED session / TYPE=VC SUBJECT=ID;
  ODS OUTPUT CovParms=CovEdPiece InfoCrit=FitEdPiece; * Save covparms, fit for comparison;
  * LSMEANS gives follow-up tests and means per group for education main effect only;
  LSMEANS EducGrp / AT (Slope12 Slope26 Age80 Reas22) = (0 0 0 0) DIFF=ALL;
  LSMEANS EducGrp / AT (Slope12 Slope26 Age80 Reas22) = (1 4 0 0) DIFF=ALL;
  * ESTIMATE statements can also give specific effects as before;
  ESTIMATE "L vs. H Educ for Intercept" EducGrp -1 0 1 ;
  ESTIMATE "M vs. H Educ for Intercept" EducGrp 0 -1 1 ;
  ESTIMATE "L vs. M Educ for Intercept" EducGrp -1 1 0 ;
  ESTIMATE "L vs. H Educ for Slope12" Slope12*EducGrp -1 0 1 ;
  ESTIMATE "M vs. H Educ for Slope12" Slope12*EducGrp 0 -1 1 ;
  ESTIMATE "L vs. M Educ for Slope12" Slope12*EducGrp -1 1 0 ;
  ESTIMATE "L vs. H Educ for Slope26" Slope26*EducGrp -1 0 1 ;
  ESTIMATE "M vs. H Educ for Slope26" Slope26*EducGrp 0 -1 1 ;
  ESTIMATE "L vs. M Educ for Slope26" Slope26*EducGrp -1 1 0 ;
```

RUN;

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	ID	233821	35419	6.60	<.0001
UN(2,1)	ID	-48793	16217	-3.01	0.0026
UN(2,2)	ID	58952	12488	4.72	<.0001
UN(3,1)	ID	-7170.04	3237.45	-2.21	0.0268
UN(3,2)	ID	-1555.12	1948.76	-0.80	0.4249
UN(3,3)	ID	2237.47	581.55	3.85	<.0001
session	ID	17673	1435.84	12.31	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8272.3	22	8316.3	8318.1	8339.6	8373.9	8395.9

Solution for Fixed Effects

Effect	Education Group (1=HS,2=BA,3=GRAD)	Estimate	Standard Error	DF	t Value	Pr > t
Intercept		1978.15	103.17	101	19.17	<.0001
Slope12		-153.14	60.7627	101	-2.52	0.0133
Slope26		-24.6403	13.0196	101	-1.89	0.0613
Age80		22.9367	8.7247	101	2.63	0.0099
Slope12*Age80		-8.9054	5.1383	101	-1.73	0.0861
Slope26*Age80		-0.5289	1.1010	101	-0.48	0.6320
Reas22		-28.5673	11.6709	101	-2.45	0.0161
Slope12*Reas22		-7.0891	6.8734	101	-1.03	0.3048
Slope26*Reas22		3.4883	1.4728	101	2.37	0.0198

EducGrp	1	-41.9718	153.41	101	-0.27	0.7850
EducGrp	2	25.4470	122.40	101	0.21	0.8357
EducGrp	3	0
Slope12*EducGrp	1	-85.9455	90.3485	101	-0.95	0.3437
Slope12*EducGrp	2	18.5834	72.0838	101	0.26	0.7971
Slope12*EducGrp	3	0
Slope26*EducGrp	1	-6.3237	19.3589	101	-0.33	0.7446
Slope26*EducGrp	2	-16.5965	15.4453	101	-1.07	0.2851
Slope26*EducGrp	3	0

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
Slope12	1	101	29.63	<.0001
Slope26	1	101	21.81	<.0001
Age80	1	101	6.91	0.0099
Slope12*Age80	1	101	3.00	0.0861
Slope26*Age80	1	101	0.23	0.6320
Reas22	1	101	5.99	0.0161
Slope12*Reas22	1	101	1.06	0.3048
Slope26*Reas22	1	101	5.61	0.0198
EducGrp	2	101	0.13	0.8774
Slope12*EducGrp	2	101	0.90	0.4104
Slope26*EducGrp	2	101	0.63	0.5347

I normally skip this box if the CLASS statement is not used, but here the last three entries give us the omnibus (df=2) tests for whether there are any education group differences on the intercept, slope12, or slope26 time slopes, not just pairwise comparisons.

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t
L vs. H Educ for Intercept	41.9718	153.41	101	0.27	0.7850
M vs. H Educ for Intercept	-25.4470	122.40	101	-0.21	0.8357
L vs. M Educ for Intercept	67.4187	132.94	101	0.51	0.6132
L vs. H Educ for Slope12	85.9455	90.3485	101	0.95	0.3437
M vs. H Educ for Slope12	-18.5834	72.0838	101	-0.26	0.7971
L vs. M Educ for Slope12	104.53	78.2936	101	1.34	0.1848
L vs. H Educ for Slope26	6.3237	19.3589	101	0.33	0.7446
M vs. H Educ for Slope26	16.5965	15.4453	101	1.07	0.2851
L vs. M Educ for Slope26	-10.2728	16.7759	101	-0.61	0.5417

Least Squares Means

Effect	Education Group (1=HS, 2=BA, 3=GRAD)	Slope12	Slope26	Age80	Reas22	Estimate	Standard Error	DF	t Value	Pr > t
EducGrp	1	0.00	0.00	0.00	0.00	1936.18	111.26	101	17.40	<.0001
EducGrp	2	0.00	0.00	0.00	0.00	2003.60	68.5957	101	29.21	<.0001
EducGrp	3	0.00	0.00	0.00	0.00	1978.15	103.17	101	19.17	<.0001
EducGrp	1	1.00	4.00	0.00	0.00	1573.24	91.9585	101	17.11	<.0001
EducGrp	2	1.00	4.00	0.00	0.00	1704.10	56.6933	101	30.06	<.0001
EducGrp	3	1.00	4.00	0.00	0.00	1726.45	85.2719	101	20.25	<.0001

Differences of Least Squares Means

Effect	Education Group (1=HS, 2=BA, 3=GRAD)	Education Group (1=HS, 2=BA, 3=GRAD)	Slope12	Slope26	Age80	Reas22	Estimate	Standard Error	DF	t Value	Pr > t
EducGrp	1	2	0.00	0.00	0.00	0.00	-67.4187	132.94	101	-0.51	0.6132
EducGrp	1	3	0.00	0.00	0.00	0.00	-41.9718	153.41	101	-0.27	0.7850
EducGrp	2	3	0.00	0.00	0.00	0.00	25.4470	122.40	101	0.21	0.8357
EducGrp	1	2	1.00	4.00	0.00	0.00	-130.86	109.87	101	-1.19	0.2365
EducGrp	1	3	1.00	4.00	0.00	0.00	-153.21	126.79	101	-1.21	0.2297
EducGrp	2	3	1.00	4.00	0.00	0.00	-22.3558	101.16	101	-0.22	0.8255

Note you must specify a value for a continuous covariate to get a predicted mean for a specific person.

Syntax and output from additional macros:

```
* Calculate difference in model fit relative to model with age and reasoning;
%FitTest(FitFewer=FitReas, NameFewer=Reasoning, FitMore=FitEdPiece, NameMore=EducGrp);
```

Add Effects of Education Group on Intercept, Slope12, and Slope26
Likelihood Ratio Test for Reasoning vs. EducGrp

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
Reasoning	8276.9	16	8308.9	8350.7	.	.	.
EducGrp	8272.3	22	8316.3	8373.9	4.54466	6	0.60339

```
* Calculate Total R2 change relative to model with age and reasoning;
%TotalR2(DV=nm3rt, PredFewer=PredReas, NameFewer=Reasoning, PredMore=PredEdPiece, NameMore=EducGrp);
```

Add Effects of Education Group on Intercept, Slope12, and Slope26
Total R2 (% Reduction) for Reasoning vs. EducGrp

Name	Pred Corr	TotalR2	Total R2Diff
Reasoning	0.40163	0.16131	.
EducGrp	0.41669	0.17363	0.012322

```
* Calculate PseudoR2 relative to model with age and reasoning;
%PseudoR2(NCov=7, CovFewer=CovReas, NameFewer=Reasoning, CovMore=CovEdPiece, NameMore=EducGrp);
```

Add Effects of Education Group on Intercept, Slope12, and Slope26
PseudoR2 (% Reduction) for Reasoning vs. EducGrp

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
Reasoning	UN(1,1)	ID	234473	35511	6.60	<.0001	.
Reasoning	UN(2,2)	ID	60504	12703	4.76	<.0001	.
Reasoning	UN(3,3)	ID	2287.42	588.37	3.89	<.0001	.
Reasoning	session	ID	17673	1435.84	12.31	<.0001	.
EducGrp	UN(1,1)	ID	233821	35419	6.60	<.0001	0.002781
EducGrp	UN(2,2)	ID	58952	12488	4.72	<.0001	0.025653
EducGrp	UN(3,3)	ID	2237.47	581.55	3.85	<.0001	0.021837
EducGrp	session	ID	17673	1435.84	12.31	<.0001	0.000000

Given that education group has no significant effects, we can drop it entirely before moving on to examine potential interactions among the time-invariant predictors of baseline age and reasoning.

Model 1d. Piecewise Model with Age*Reasoning on Intercept, Slope12, Slope26

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i} (\text{Slope12}_{ti}) + \beta_{2i} (\text{Slope26}_{ti}) + e_{ti}$

Level 2:

Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01} (\text{Age}_i - 80) + \gamma_{02} (\text{Reason}_i - 22) + \gamma_{03} (\text{Age}_i - 80)(\text{Reason}_i - 22) + U_{0i}$

Slope12: $\beta_{1i} = \gamma_{10} + \gamma_{11} (\text{Age}_i - 80) + \gamma_{12} (\text{Reason}_i - 22) + \gamma_{13} (\text{Age}_i - 80)(\text{Reason}_i - 22) + U_{1i}$

Slope26: $\beta_{2i} = \gamma_{20} + \gamma_{21} (\text{Age}_i - 80) + \gamma_{22} (\text{Reason}_i - 22) + \gamma_{23} (\text{Age}_i - 80)(\text{Reason}_i - 22) + U_{2i}$

```

TITLE1 "Add Fixed Effects for Age on Intercept, Slope12, and Slope26";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST NAMELEN=100 IC METHOD=ML;
  CLASS ID session;
  MODEL nm3rt = Slope12 Slope26 Age80 Age80*Slope12 Age80*Slope26 Reas22 Reas22*Slope12
              Reas22*Slope26 Age80*Reas22 Age80*Reas22*Slope12 Age80*Reas22*Slope26
              / SOLUTION DDFM=Satterthwaite OUTPM=PredAgeReas; * Save fixed-predicted RT;
  RANDOM INTERCEPT Slope12 Slope26 / G TYPE=UN SUBJECT=ID;
  REPEATED session / TYPE=VC SUBJECT=ID;
  ODS OUTPUT CovParms=CovAgeReas InfoCrit=FitAgeReas; * Save covparms, fit for comparison;
* Age simple effects to decompose interactions;
ESTIMATE "Age Effect on Intercept, Reas 17" Age80 1 Reas22*Age80 -5;
ESTIMATE "Age Effect on Intercept, Reas 22" Age80 1 Reas22*Age80 0;
ESTIMATE "Age Effect on Intercept, Reas 27" Age80 1 Reas22*Age80 5;

ESTIMATE "Age Effect on Slope12, Reas 17" Age80*Slope12 1 Reas22*Age80*Slope12 -5;
ESTIMATE "Age Effect on Slope12, Reas 22" Age80*Slope12 1 Reas22*Age80*Slope12 0;
ESTIMATE "Age Effect on Slope12, Reas 27" Age80*Slope12 1 Reas22*Age80*Slope12 5;
ESTIMATE "Age Effect on Slope26, Reas 17" Age80*Slope26 1 Reas22*Age80*Slope26 -5;
ESTIMATE "Age Effect on Slope26, Reas 22" Age80*Slope26 1 Reas22*Age80*Slope26 0;
ESTIMATE "Age Effect on Slope26, Reas 27" Age80*Slope26 1 Reas22*Age80*Slope26 5;

* Reasoning simple effects to decompose interactions;
ESTIMATE "Reasoning Effect on Intercept, Age 70" Reas22 1 Reas22*Age80 -10;
ESTIMATE "Reasoning Effect on Intercept, Age 80" Reas22 1 Reas22*Age80 0;
ESTIMATE "Reasoning Effect on Intercept, Age 90" Reas22 1 Reas22*Age80 10;

ESTIMATE "Reasoning Effect on Slope12, Age 70" Reas22*Slope12 1 Reas22*Age80*Slope12 -10;
ESTIMATE "Reasoning Effect on Slope12, Age 80" Reas22*Slope12 1 Reas22*Age80*Slope12 0;
ESTIMATE "Reasoning Effect on Slope12, Age 90" Reas22*Slope12 1 Reas22*Age80*Slope12 10;
ESTIMATE "Reasoning Effect on Slope26, Age 70" Reas22*Slope26 1 Reas22*Age80*Slope26 -10;
ESTIMATE "Reasoning Effect on Slope26, Age 80" Reas22*Slope26 1 Reas22*Age80*Slope26 0;
ESTIMATE "Reasoning Effect on Slope26, Age 90" Reas22*Slope26 1 Reas22*Age80*Slope26 10;
RUN;

```

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	ID	233821	35419	6.60	<.0001
UN(2,1)	ID	-46952	16176	-2.90	0.0037
UN(2,2)	ID	59370	12546	4.73	<.0001
UN(3,1)	ID	-7216.10	3254.58	-2.22	0.0266
UN(3,2)	ID	-1780.00	1968.22	-0.90	0.3658
UN(3,3)	ID	2279.52	587.29	3.88	<.0001
session	ID	17673	1435.84	12.31	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8274.9	19	8312.9	8314.2	8333.0	8362.5	8381.5

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	1974.57	52.7612	101	37.42	<.0001
Slope12	-151.52	31.1471	101	-4.86	<.0001
Slope26	-34.1783	6.6928	101	-5.11	<.0001
Age80	22.7598	8.7330	101	2.61	0.0105
Slope12*Age80	-8.4366	5.1555	101	-1.64	0.1049
Slope26*Age80	-0.4866	1.1078	101	-0.44	0.6614
Reas22	-28.0448	11.4108	101	-2.46	0.0157
Slope12*Reas22	-3.4941	6.7363	101	-0.52	0.6051
Slope26*Reas22	3.4494	1.4475	101	2.38	0.0190
Age80*Reas22	-0.9317	1.8208	101	-0.51	0.6100
Slope12*Age80*Reas22	1.2290	1.0749	101	1.14	0.2556
Slope26*Age80*Reas22	0.1026	0.2310	101	0.44	0.6579

Label	Estimates				
	Estimate	Standard Error	DF	t Value	Pr > t
Age Effect on Intercept, Reas 17	27.4184	12.2659	101	2.24	0.0276
Age Effect on Intercept, Reas 22	22.7598	8.7330	101	2.61	0.0105
Age Effect on Intercept, Reas 27	18.1011	12.9553	101	1.40	0.1654
Age Effect on Slope12, Reas 17	-14.5818	7.2411	101	-2.01	0.0467
Age Effect on Slope12, Reas 22	-8.4366	5.1555	101	-1.64	0.1049
Age Effect on Slope12, Reas 27	-2.2914	7.6481	101	-0.30	0.7651
Age Effect on Slope26, Reas 17	-0.9994	1.5559	101	-0.64	0.5221
Age Effect on Slope26, Reas 22	-0.4866	1.1078	101	-0.44	0.6614
Age Effect on Slope26, Reas 27	0.02627	1.6434	101	0.02	0.9873
Reasoning Effect on Intercept, Age 70	-18.7275	19.8977	101	-0.94	0.3489
Reasoning Effect on Intercept, Age 80	-28.0448	11.4108	101	-2.46	0.0157
Reasoning Effect on Intercept, Age 90	-37.3622	22.9683	101	-1.63	0.1069
Reasoning Effect on Slope12, Age 70	-15.7845	11.7464	101	-1.34	0.1820
Reasoning Effect on Slope12, Age 80	-3.4941	6.7363	101	-0.52	0.6051
Reasoning Effect on Slope12, Age 90	8.7963	13.5592	101	0.65	0.5180
Reasoning Effect on Slope26, Age 70	2.4237	2.5240	101	0.96	0.3392
Reasoning Effect on Slope26, Age 80	3.4494	1.4475	101	2.38	0.0190
Reasoning Effect on Slope26, Age 90	4.4751	2.9136	101	1.54	0.1277

Which fixed effects are now conditional on age?

Which fixed effects are now conditional on reasoning?

Syntax and output from additional macros:

```
* Calculate difference in model fit relative to age and reasoning main effects model;
%FitTest(FitFewer=FitReas, NameFewer=Age+Reasoning, FitMore=FitAgeReas, NameMore=Age*Reasoning);
```

Name	Neg2Log						
	Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
Age+Reasoning	8276.9	16	8308.9	8350.7	.	.	.
Age*Reasoning	8274.9	19	8312.9	8362.5	2.03841	3	0.56447

```
* Calculate Total R2 change relative to age and reasoning main effects model;
%TotalR2(DV=nm3rt, PredFewer=PredReas, NameFewer=Age+Reasoning, PredMore=PredAgeReas, NameMore=Age*Reasoning);
```

Name	Pred		Total	
	Corr	TotalR2	R2Diff	
Age+Reasoning	0.40163	0.16131	.	
Age*Reasoning	0.40306	0.16246	.001148258	

```
* Calculate PseudoR2 relative to age and reasoning main effects model;
%PseudoR2(NCov=7, CovFewer=CovReas, NameFewer=Age+Reasoning, CovMore=CovAgeReas, NameMore=Age*Reasoning);
```

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
Age+Reasoning	UN(1,1)	ID	234473	35511	6.60	<.0001	.
Age+Reasoning	UN(2,2)	ID	60504	12703	4.76	<.0001	.
Age+Reasoning	UN(3,3)	ID	2287.42	588.37	3.89	<.0001	.
Age+Reasoning	session	ID	17673	1435.84	12.31	<.0001	.
Age*Reasoning	UN(1,1)	ID	233821	35419	6.60	<.0001	0.002781
Age*Reasoning	UN(2,2)	ID	59370	12546	4.73	<.0001	0.018752
Age*Reasoning	UN(3,3)	ID	2279.52	587.29	3.88	<.0001	0.003454
Age*Reasoning	session	ID	17673	1435.84	12.31	<.0001	0.000000

Based on the lack of significance of the higher-order interactions, I'd say we're done with this model. Age and reasoning as main effects in predicting intercept, slope12, and slope26 seems to be the best model.

Model 2a. Quadratic Model with Age, Reasoning on Intercept, Linear, Quadratic Time Slopes

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}(\text{Session}_{ti} - 1) + \beta_{2i}(\text{Session}_{ti} - 1)^2 + e_{ti}$$

Level 2:

$$\text{Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01}(\text{Age}_i - 80) + \gamma_{02}(\text{Reason}_i - 22) + U_{0i}$$

$$\text{Linear: } \beta_{1i} = \gamma_{10} + \gamma_{11}(\text{Age}_i - 80) + \gamma_{12}(\text{Reason}_i - 22) + U_{1i}$$

$$\text{Quadratic: } \beta_{2i} = \gamma_{20} + \gamma_{21}(\text{Age}_i - 80) + \gamma_{22}(\text{Reason}_i - 22) + U_{2i}$$

```
TITLE1 "Age + Reasoning on Intercept, Linear, and Quadratic Time Slopes";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST IC NAMELEN=100 METHOD=ML;
  CLASS ID session;
  MODEL nm3rt = c1sess c1sess*c1sess Age80 c1sess*Age80 c1sess*c1sess*Age80
               Reas22 c1sess*Reas22 c1sess*c1sess*Reas22
               / SOLUTION DDFM=Satterthwaite OUTPM=PredARQuad; * Save fixed-predicted RT;
  RANDOM INTERCEPT c1sess c1sess*c1sess / G TYPE=UN SUBJECT=ID;
  REPEATED session / TYPE=VC SUBJECT=ID;
  ODS OUTPUT CovParms=CovARQuad InfoCrit=FitARQuad; * Save covparms, fit for comparison;
* Requesting additional effects for age;
ESTIMATE "Age Effect at Session 1" Age80 1 c1sess*Age80 0 c1sess*c1sess*Age80 0;
ESTIMATE "Age Effect at Session 2" Age80 1 c1sess*Age80 1 c1sess*c1sess*Age80 1;
ESTIMATE "Age Effect at Session 3" Age80 1 c1sess*Age80 2 c1sess*c1sess*Age80 4;
ESTIMATE "Age Effect at Session 4" Age80 1 c1sess*Age80 3 c1sess*c1sess*Age80 9;
ESTIMATE "Age Effect at Session 5" Age80 1 c1sess*Age80 4 c1sess*c1sess*Age80 16;
ESTIMATE "Age Effect at Session 6" Age80 1 c1sess*Age80 5 c1sess*c1sess*Age80 25;
ESTIMATE "Age*Linear Time Slope at Session 1" c1sess*Age80 1 c1sess*c1sess*Age80 0;
ESTIMATE "Age*Linear Time Slope at Session 2" c1sess*Age80 1 c1sess*c1sess*Age80 2;
ESTIMATE "Age*Linear Time Slope at Session 3" c1sess*Age80 1 c1sess*c1sess*Age80 4;
ESTIMATE "Age*Linear Time Slope at Session 4" c1sess*Age80 1 c1sess*c1sess*Age80 6;
ESTIMATE "Age*Linear Time Slope at Session 5" c1sess*Age80 1 c1sess*c1sess*Age80 8;
ESTIMATE "Age*Linear Time Slope at Session 6" c1sess*Age80 1 c1sess*c1sess*Age80 10;
* Requesting additional effects for reasoning;
ESTIMATE "Reasoning Effect at Session 1" Reas22 1 c1sess*Reas22 0 c1sess*c1sess*Reas22 0;
ESTIMATE "Reasoning Effect at Session 2" Reas22 1 c1sess*Reas22 1 c1sess*c1sess*Reas22 1;
ESTIMATE "Reasoning Effect at Session 3" Reas22 1 c1sess*Reas22 2 c1sess*c1sess*Reas22 4;
ESTIMATE "Reasoning Effect at Session 4" Reas22 1 c1sess*Reas22 3 c1sess*c1sess*Reas22 9;
ESTIMATE "Reasoning Effect at Session 5" Reas22 1 c1sess*Reas22 4 c1sess*c1sess*Reas22 16;
ESTIMATE "Reasoning Effect at Session 6" Reas22 1 c1sess*Reas22 5 c1sess*c1sess*Reas22 25;
ESTIMATE "Reasoning*Linear Time Slope at Session 1" c1sess*Reas22 1 c1sess*c1sess*Reas22 0;
ESTIMATE "Reasoning*Linear Time Slope at Session 2" c1sess*Reas22 1 c1sess*c1sess*Reas22 2;
ESTIMATE "Reasoning*Linear Time Slope at Session 3" c1sess*Reas22 1 c1sess*c1sess*Reas22 4;
ESTIMATE "Reasoning*Linear Time Slope at Session 4" c1sess*Reas22 1 c1sess*c1sess*Reas22 6;
ESTIMATE "Reasoning*Linear Time Slope at Session 5" c1sess*Reas22 1 c1sess*c1sess*Reas22 8;
ESTIMATE "Reasoning*Linear Time Slope at Session 6" c1sess*Reas22 1 c1sess*c1sess*Reas22 10;
RUN;
```

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	ID	228049	34464	6.62	<.0001
UN(2,1)	ID	-31230	10649	-2.93	0.0034
UN(2,2)	ID	24041	5588.98	4.30	<.0001
UN(3,1)	ID	3748.22	1746.63	2.15	0.0319
UN(3,2)	ID	-3618.98	937.05	-3.86	0.0001
UN(3,3)	ID	580.07	164.19	3.53	0.0002
session	ID	20298	1649.11	12.31	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8297.7	16	8329.7	8330.7	8346.7	8371.6	8387.6

Solution for Fixed Effects
Standard

Effect	Estimate	Error	DF	t Value	Pr > t
Intercept	1966.47	49.6658	101	39.59	<.0001
c1sess	-119.74	19.7742	101	-6.06	<.0001
c1sess*c1sess	13.3036	3.3656	101	3.95	0.0001
Age80	22.2782	8.6018	101	2.59	0.0110
c1sess*Age80	-6.4921	3.4247	101	-1.90	0.0609
c1sess*c1sess*Age80	0.9601	0.5829	101	1.65	0.1026
Reas22	-27.1004	11.1141	101	-2.44	0.0165
c1sess*Reas22	-3.5917	4.4250	101	-0.81	0.4189
c1sess*c1sess*Reas22	1.1575	0.7531	101	1.54	0.1274

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t
Age Effect at Session 1	22.2782	8.6018	101	2.59	0.0110
Age Effect at Session 2	16.7462	7.7118	101	2.17	0.0322
Age Effect at Session 3	13.1346	7.5432	101	1.74	0.0847
Age Effect at Session 4	11.4432	7.4506	101	1.54	0.1277
Age Effect at Session 5	11.6721	7.2596	101	1.61	0.1110
Age Effect at Session 6	13.8212	7.3495	101	1.88	0.0629
Age*Linear Time Slope at Session 1	-6.4921	3.4247	101	-1.90	0.0609
Age*Linear Time Slope at Session 2	-4.5718	2.3288	101	-1.96	0.0524
Age*Linear Time Slope at Session 3	-2.6515	1.3551	101	-1.96	0.0531
Age*Linear Time Slope at Session 4	-0.7313	0.9835	101	-0.74	0.4589
Age*Linear Time Slope at Session 5	1.1890	1.6781	101	0.71	0.4802
Age*Linear Time Slope at Session 6	3.1093	2.7172	101	1.14	0.2552
Reasoning Effect at Session 1	-27.1004	11.1141	101	-2.44	0.0165
Reasoning Effect at Session 2	-29.5346	9.9642	101	-2.96	0.0038
Reasoning Effect at Session 3	-29.6537	9.7464	101	-3.04	0.0030
Reasoning Effect at Session 4	-27.4578	9.6268	101	-2.85	0.0053
Reasoning Effect at Session 5	-22.9468	9.3799	101	-2.45	0.0162
Reasoning Effect at Session 6	-16.1207	9.4960	101	-1.70	0.0927
Reasoning*Linear Time Slope at Session 1	-3.5917	4.4250	101	-0.81	0.4189
Reasoning*Linear Time Slope at Session 2	-1.2767	3.0090	101	-0.42	0.6723
Reasoning*Linear Time Slope at Session 3	1.0384	1.7509	101	0.59	0.5545
Reasoning*Linear Time Slope at Session 4	3.3535	1.2707	101	2.64	0.0096
Reasoning*Linear Time Slope at Session 5	5.6686	2.1683	101	2.61	0.0103
Reasoning*Linear Time Slope at Session 6	7.9836	3.5109	101	2.27	0.0251

From these results *it appears* we could remove both the interaction of reasoning with both the linear and quadratic time slopes, but keep in mind how correlated those terms are... let's see what happens if we just remove just the reasoning*quadratic time interaction for now.

Model 2b. Quadratic Model without Reasoning by Quadratic Time Slope

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i} (\text{Session}_{ti} - 1) + \beta_{2i} (\text{Session}_{ti} - 1)^2 + e_{ti}$

Level 2:

Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01} (\text{Age}_i - 80) + \gamma_{02} (\text{Reason}_i - 22) + U_{0i}$

Linear: $\beta_{1i} = \gamma_{10} + \gamma_{11} (\text{Age}_i - 80) + \gamma_{12} (\text{Reason}_i - 22) + U_{1i}$

Quadratic: $\beta_{2i} = \gamma_{20} + \gamma_{21} (\text{Age}_i - 80) + U_{2i}$

```
TITLE1 "Remove Reasoning Effect on Quadratic Time Slope";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST IC NAMELEN=100 METHOD=ML;
  CLASS ID session;
  MODEL nm3rt = c1sess c1sess*c1sess Age80 c1sess*Age80 c1sess*c1sess*Age80 Reas22
  c1sess*Reas22 / SOLUTION DDFM=Satterthwaite OUTPM=PredARLin; * Save fixed-predicted RT;
  RANDOM INTERCEPT c1sess c1sess*c1sess / G TYPE=UN SUBJECT=ID;
  REPEATED session / TYPE=VC SUBJECT=ID;
  ODS OUTPUT CovParms=CovARLin InfoCrit=FitARQLin; * Save covparms, fit for comparison;
  * Requesting additional effects for age - code is exactly the same;
  * Requesting additional effects for reasoning;
  ESTIMATE "Reasoning Effect at Session 1" Reas22 1 c1sess*Reas22 0;
  ESTIMATE "Reasoning Effect at Session 2" Reas22 1 c1sess*Reas22 1;
  ESTIMATE "Reasoning Effect at Session 3" Reas22 1 c1sess*Reas22 2;
  ESTIMATE "Reasoning Effect at Session 4" Reas22 1 c1sess*Reas22 3;
  ESTIMATE "Reasoning Effect at Session 5" Reas22 1 c1sess*Reas22 4;
  ESTIMATE "Reasoning Effect at Session 6" Reas22 1 c1sess*Reas22 5; RUN;
```

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	ID	228688	34635	6.60	<.0001
UN(2,1)	ID	-31959	10877	-2.94	0.0033
UN(2,2)	ID	24872	5711.58	4.35	<.0001
UN(3,1)	ID	3877.66	1786.86	2.17	0.0300
UN(3,2)	ID	-3766.27	957.50	-3.93	<.0001
UN(3,3)	ID	606.16	167.70	3.61	0.0002
session	ID	20298	1649.11	12.31	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8300.1	15	8330.1	8330.9	8345.9	8369.3	8384.3

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	1969.80	49.6821	101	39.65	<.0001
c1sess	-123.54	19.8277	101	-6.23	<.0001
c1sess*c1sess	13.9774	3.3754	101	4.14	<.0001
Age80	20.8470	8.5613	103	2.44	0.0166
c1sess*Age80	-4.8610	3.2905	102	-1.48	0.1427
c1sess*c1sess*Age80	0.6709	0.5580	101	1.20	0.2321
Reas22	-32.8284	10.4706	101	-3.14	0.0022
c1sess*Reas22	2.9363	1.2412	101	2.37	0.0199 → Different result!

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t
Reasoning Effect at Session 1	-32.8284	10.4706	101	-3.14	0.0022
Reasoning Effect at Session 2	-29.8921	9.9615	101	-3.00	0.0034
Reasoning Effect at Session 3	-26.9558	9.5870	101	-2.81	0.0059
Reasoning Effect at Session 4	-24.0195	9.3632	101	-2.57	0.0118
Reasoning Effect at Session 5	-21.0831	9.3012	101	-2.27	0.0255
Reasoning Effect at Session 6	-18.1468	9.4040	101	-1.93	0.0564

```
* Calculate difference in model fit relative to model with reasoning*quad;
%FitTest(FitFewer=FitARQLin, NameFewer=NOReasoning*Quad, FitMore=FitARQuad,
NameMore=YESReasoning*Quad);
```

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
NOReasoning*Quad	8300.1	15	8330.1	8369.3	.	.	.
YESReasoning*Quad	8297.7	16	8329.7	8371.6	2.33501	1	0.12649

Model 2c. Quadratic Model adding Effects of Education Group on Intercept, Linear, Quadratic Time

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Session}_{ti} - 1) + \beta_{2i}(\text{Session}_{ti} - 1)^2 + e_{ti}$

Level 2:

Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01}(\text{Age}_i - 80) + \gamma_{02}(\text{Reason}_i - 22) + \gamma_{03}(\text{Highvs.LowEd}_i) + \gamma_{04}(\text{Highvs.MedEd}_i) + U_{0i}$

Linear: $\beta_{1i} = \gamma_{10} + \gamma_{11}(\text{Age}_i - 80) + \gamma_{12}(\text{Reason}_i - 22) + \gamma_{13}(\text{Highvs.LowEd}_i) + \gamma_{14}(\text{Highvs.MedEd}_i) + U_{1i}$

Quadratic: $\beta_{2i} = \gamma_{20} + \gamma_{21}(\text{Age}_i - 80) + \gamma_{23}(\text{Highvs.LowEd}_i) + \gamma_{24}(\text{Highvs.MedEd}_i) + U_{2i}$

```
TITLE1 "Add Effect of Education Group on Intercept, Linear, and Quadratic";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST IC NAMELEN=100 METHOD=ML;
  CLASS ID EducGrp session;
  MODEL nm3rt = c1sess c1sess*c1sess Age80 c1sess*Age80 c1sess*c1sess*Age80
               Reas22 c1sess*Reas22 EducGrp c1sess*EducGrp c1sess*c1sess*EducGrp
               / SOLUTION DDFM=Satterthwaite OUTPM=PredEdQuad; * Save fixed-predicted RT;
  RANDOM INTERCEPT c1sess c1sess*c1sess / G TYPE=UN SUBJECT=ID;
  REPEATED session / TYPE=VC SUBJECT=ID;
  ODS OUTPUT CovParms=CovEdQuad InfoCrit=FitEdQuad; * Save covparms, fit for comparison;

  * LSMEANS gives follow-up tests and means per group for education main effect only;
  LSMEANS EducGrp / AT (c1sess Age80 Reas22) = (0 0 0) DIFF=ALL;
  LSMEANS EducGrp / AT (c1sess Age80 Reas22) = (5 0 0) DIFF=ALL;

  * ESTIMATE statements can also give specific effects as before;
  ESTIMATE "L vs. H Educ for Intercept"      EducGrp -1 0 1 ;
  ESTIMATE "M vs. H Educ for Intercept"      EducGrp 0 -1 1 ;
  ESTIMATE "L vs. M Educ for Intercept"      EducGrp -1 1 0 ;

  ESTIMATE "L vs. H Educ for Linear Time"    c1sess*EducGrp -1 0 1 ;
  ESTIMATE "M vs. H Educ for Linear Time"    c1sess*EducGrp 0 -1 1 ;
  ESTIMATE "L vs. M Educ for Linear Time"    c1sess*EducGrp -1 1 0 ;

  ESTIMATE "L vs. H Educ for Quadratic Time" c1sess*c1sess*EducGrp -1 0 1 ;
  ESTIMATE "M vs. H Educ for Quadratic Time" c1sess*c1sess*EducGrp 0 -1 1 ;
  ESTIMATE "L vs. M Educ for Quadratic Time" c1sess*c1sess*EducGrp -1 1 0 ;
```

RUN;

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	ID	228585	34693	6.59	<.0001
UN(2,1)	ID	-33273	10909	-3.05	0.0023
UN(2,2)	ID	24129	5614.80	4.30	<.0001
UN(3,1)	ID	4125.93	1788.44	2.31	0.0211
UN(3,2)	ID	-3633.38	939.20	-3.87	0.0001
UN(3,3)	ID	581.50	164.31	3.54	0.0002
session	ID	20298	1649.11	12.31	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8295.4	21	8337.4	8338.9	8359.6	8392.3	8413.3

Solution for Fixed Effects

Effect	Education Group (1=HS,2=BA,3=GRAD)	Estimate	Standard Error	DF	t Value	Pr > t
Intercept		1961.89	101.79	100	19.27	<.0001
c1sess		-106.50	40.2761	101	-2.64	0.0095
c1sess*c1sess		12.4797	6.8474	101	1.82	0.0713
Age80		20.2893	8.5600	102	2.37	0.0196
c1sess*Age80		-4.5758	3.2667	102	-1.40	0.1643
c1sess*c1sess*Age80		0.6177	0.5533	101	1.12	0.2669
Reas22		-36.6228	10.7638	101	-3.40	0.0010
c1sess*Reas22		2.9788	1.2799	101	2.33	0.0219
EducGrp	1	-51.3811	151.06	101	-0.34	0.7345
EducGrp	2	37.6427	120.87	100	0.31	0.7561
EducGrp	3	0
c1sess*EducGrp	1	-70.2445	59.0672	101	-1.19	0.2371
c1sess*EducGrp	2	-4.3577	48.1238	100	-0.09	0.9280
c1sess*EducGrp	3	0
c1sess*c1sess*EducGrp	1	11.0653	10.0300	101	1.10	0.2726
c1sess*c1sess*EducGrp	2	-1.4641	8.1865	101	-0.18	0.8584
c1sess*c1sess*EducGrp	3	0

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
c1sess	1	101	37.28	<.0001
c1sess*c1sess	1	101	18.35	<.0001
Age80	1	102	5.62	0.0196
c1sess*Age80	1	102	1.96	0.1643
c1sess*c1sess*Age80	1	101	1.25	0.2669
Reas22	1	101	11.58	0.0010
c1sess*Reas22	1	101	5.42	0.0219
EducGrp	2	101	0.24	0.7874
c1sess*EducGrp	2	101	0.96	0.3860
c1sess*c1sess*EducGrp	2	101	1.09	0.3395

I normally skip this box if the CLASS statement is not used, but here the last three entries give us the omnibus (df=2) tests for whether there are any education group differences on the intercept, linear, or quadratic time slopes, not just pairwise comparisons.

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t
L vs. H Educ for Intercept	51.3811	151.06	101	0.34	0.7345
M vs. H Educ for Intercept	-37.6427	120.87	100	-0.31	0.7561
L vs. M Educ for Intercept	89.0238	130.74	102	0.68	0.4975
L vs. H Educ for Linear Time	70.2445	59.0672	101	1.19	0.2371
M vs. H Educ for Linear Time	4.3577	48.1238	100	0.09	0.9280
L vs. M Educ for Linear Time	65.8868	50.7047	101	1.30	0.1967
L vs. H Educ for Quadratic Time	-11.0653	10.0300	101	-1.10	0.2726
M vs. H Educ for Quadratic Time	1.4641	8.1865	101	0.18	0.8584
L vs. M Educ for Quadratic Time	-12.5294	8.6028	101	-1.46	0.1484

Least Squares Means

Effect	Education Group (1=HS, 2=BA, 3=GRAD)	c1sess	Age80	Reas22	Estimate	Standard Error	DF	t Value	Pr > t
EducGrp	1	0.00	0.00	0.00	1910.51	109.66	101	17.42	<.0001
EducGrp	2	0.00	0.00	0.00	1999.53	67.5565	101	29.60	<.0001
EducGrp	3	0.00	0.00	0.00	1961.89	101.79	100	19.27	<.0001
EducGrp	1	5.00	0.00	0.00	1615.41	93.3431	101	17.31	<.0001
EducGrp	2	5.00	0.00	0.00	1720.63	57.5381	101	29.90	<.0001
EducGrp	3	5.00	0.00	0.00	1741.38	86.5736	101	20.11	<.0001

Note you must specify a value for a continuous covariate to get a predicted mean for a specific person.

Differences of Least Squares Means

Effect	Education Group		c1sess	Age80	Reas22	Estimate	Standard Error	DF	t Value	Pr > t
	(1=HS, 2=BA, 3=GRAD)	(1=HS, 2=BA, 3=GRAD)								
EducGrp	1	2	0.00	0.00	0.00	-89.0238	130.74	102	-0.68	0.4975
EducGrp	1	3	0.00	0.00	0.00	-51.3811	151.06	101	-0.34	0.7345
EducGrp	2	3	0.00	0.00	0.00	37.6427	120.87	100	0.31	0.7561
EducGrp	1	2	5.00	0.00	0.00	-105.22	111.48	101	-0.94	0.3475
EducGrp	1	3	5.00	0.00	0.00	-125.97	128.68	101	-0.98	0.3299
EducGrp	2	3	5.00	0.00	0.00	-20.7487	102.72	101	-0.20	0.8403

```
* Calculate difference in model fit relative to model with reasoning*linear only;
%FitTest(FitFewer=FitARQLin, NameFewer=NOReasoning*Quad, FitMore=FitEdQuad, NameMore=EducGrp);
```

Name	Neg2Log		AIC	BIC	DevDiff	DFdiff	Pvalue
	Like	Parms					
NOReasoning*Quad	8300.1	15	8330.1	8369.3	.	.	.
EducGrp	8295.4	21	8337.4	8392.3	4.69813	6	0.58307

```
* Calculate Total R2 change relative to model with reasoning*linear only;
%TotalR2(DV=nm3rt, PredFewer=PredARLin, NameFewer=NOReasoning*Quad, PredMore=PredEdQuad, NameMore=EducGrp);
```

Add Effect of Education Group on Intercept, Linear, and Quadratic
Total R2 (% Reduction) for NOReasoning*Quad vs. EducGrp

Name	Pred	TotalR2	Total
	Corr		R2Diff
NOReasoning*Quad	0.40008	0.16006	.
EducGrp	0.41510	0.17231	0.012242

```
* Calculate PseudoR2 relative to model with reasoning*linear only;
%PseudoR2(NCov=7, CovFewer=CovARLin, NameFewer=NOReasoning*Quad, CovMore=CovEdQuad, NameMore=EducGrp);
```

Add Effect of Education Group on Intercept, Linear, and Quadratic
PseudoR2 (% Reduction) for NOReasoning*Quad vs. EducGrp

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
NOReasoning*Quad	UN(1,1)	ID	228688	34635	6.60	<.0001	.
NOReasoning*Quad	UN(2,2)	ID	24872	5711.58	4.35	<.0001	.
NOReasoning*Quad	UN(3,3)	ID	606.16	167.70	3.61	0.0002	.
NOReasoning*Quad	session	ID	20298	1649.11	12.31	<.0001	.
EducGrp	UN(1,1)	ID	228585	34693	6.59	<.0001	0.000451
EducGrp	UN(2,2)	ID	24129	5614.80	4.30	<.0001	0.029842
EducGrp	UN(3,3)	ID	581.50	164.31	3.54	0.0002	0.040685
EducGrp	session	ID	20298	1649.11	12.31	<.0001	-0.000000

Based on the lack of significance of the effect of education, I'd say we're done with this model (I had previously tried age*reasoning, and none of those higher-order effects were significant). The age*quadratic interaction could probably be removed, but I choose to leave it in as a control.